

# LINEAR CURVE FITTING

by M. Kostic 9/93,6/94

First, read the data from external files: DATA\_xy.PRN, or DATAx.PRN; or DATAx.PRN and DATAy.PRN, i.e.:

```
xy := READPRN("Data_xy")
```

**NOTE:** In MathCAD Ver.7 it should be:

```
xy := READPRN("drive:\path\DATA_xy.prn")
```

If data are in two column-matrix, then, separate the **x** (0th) and the **y** (1st) columns in separate vectors **x** and **y** (Remember that a matrix subscript starts from 0 in MathCAD):

```
x := xy<0>      y := xy<1>
```

or define data vectors **x** and **y**:

$$x := \begin{pmatrix} 1.4 \\ 4.7 \\ 17.3 \\ 82.9 \\ 171.6 \\ 1227.1 \end{pmatrix} \quad y := \begin{pmatrix} 0.5 \\ 1.1 \\ 2.0 \\ 2.9 \\ 5.1 \\ 10.0 \end{pmatrix}$$

Evaluate the number of data points, or the size-length of vector **x** or **y**:

```
N := length(x)      N = 6
```

Check (see) couple of data

Then, data range variable will be:

$$x_0 = 1.4$$

$$x_{N-1} = 1.227 \times 10^3$$

```
i := 0..N - 1
```

$$y_0 = 0.5$$

$$y_{N-1} = 10$$

Compute sample statistics using the built-in MathCAD functions:

$$\text{mean}(x) = 250.833$$

$$\text{mean}(y) = 3.6$$

$$\text{var}(x) = 1.942 \times 10^5$$

$$\text{var}(y) = 10.353$$

$$\text{stdev}(x) = 440.627$$

$$\text{stdev}(y) = 3.218$$

$$\text{corr}(x, y) = 0.941$$

$$bl := \text{intercept}(x, y)$$


$$bl = 1.875$$

$$ml := \text{slope}(x, y)$$

$$ml = 6.875 \times 10^{-3}$$

$$\text{linear}(x) := bl + ml \cdot x$$

Finally, plot the data and the fitted line:



# CURVE FITTING with POLYNOMIAL of m-ORDER by M. Kostic 9/93, 6/95

**NOTE:** The nomenclature corresponds to "Theory and Design for Mechanical Measurements, 2nd Ed." by Figliola and Beasley, p.149, unless given otherwise.

For a polynomial of **m**-order ( $m=0,1,2,\dots,N-1$ , may be **arbitrary** integer within this range):

$$\underline{m} := 2 \quad \dots \text{for example}$$

Then, the range variable and the degree of freedom will be:

$$j := 0..m \quad v_m := m + 1 \quad v := N - v_m$$

For a polynomial curve fit:

$$y_c = a_0 + a_1 \cdot x + a_2 \cdot x^2 + \dots + a_m \cdot x^m = \sum_{i=0}^m (a_i \cdot x^i)$$

and given data set  $\{(x_i, y_i), i=0,1,\dots,N-1\}$  of **N** data points, we may determine coefficients  $a_i$  which will provide the best fit for the given conditions, using the least-square method.

## Curve fit using matrix operations

Create **X** matrix ( [N X (m+1)] order) first:  $X^{(j)} := x^j$

and calculate (effectively) the polynomial coefficients:

$$a := (X^T \cdot X)^{-1} \cdot (X^T \cdot y) \quad a = \begin{pmatrix} 0.945 \\ 0.027 \\ -1.584 \times 10^{-5} \end{pmatrix} \quad \dots \text{the polynomial coefficients } a_i$$

**...HOW SIMPLE!!**

For "zero-order" polynomial we need a little "fix-up"

$$a_0 := \text{if}(m = 0, a, a_0)$$

Fitted curve ( $j=0\dots m$ ) is:

$$y_{c_i} := \sum_j [a_j \cdot (x_i)^j]$$

Compute the *standard error of the fit* **Sxy**:

$$d := y - y_c \quad D := \sum d^2 \quad D_0 := \sum (y - \text{mean}(y))^2$$

$$S_{xy} := \sqrt{\frac{D}{N - v_m}} \quad S_{x0} := \sqrt{\frac{D_0}{N - 1}}$$

$$S_{xy} = 0.453$$

$$S_{x0} = 3.525$$

is the Reference standard error for the "zero-order" fit, **Sy** in our Text

Correlation coefficient:

$$r := \sqrt{1 - \left(\frac{S_{xy}}{S_{x0}}\right)^2} \quad r = 0.992$$

Now plot the data and the fitted curve to visualize the results :

by M. Kostic 9/93, 6/95

If the number of data is smaller than, say  $N_c=60$ , plot the polynomial curve with at least **Nc** data

$N_c := 60$

$N = 6$  ...number of data points

$N_c := \text{if}(N < N_c, N_c, N)$

$$dc := \frac{x_{N-1} - x_0}{N_c - 1}$$

$N_c = 60$  ...number of points to plot the curve

$ic := 0..N_c - 1$

$m = 2$  ...order of polynomial


$xc_{ic} := x_0 + dc \cdot ic$

Check that these two are the same:

$$yc_{ic} := \sum_j \left[ a_j \cdot (xc_{ic})^j \right]$$

$$x_{N-1} = 1.227 \times 10^3$$

$$xc_{N_c-1} = 1.227 \times 10^3$$



DATAxy = "Data-PRN\DataxyPoli.prn"